AD-781 649

VOLUME INTERSECTION OF TWO IDENTICAL RIGHT CIRCULAR CONES

R. G. Hughes, et al

Naval Research Laboratory
Washington, D. C.

June 1974

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

AD 781 649

REPORT DOCUMENTATION PAGE

READ INSTRUCTIONS BEFORE COMPLETING FORM

1. REPORT NUMBER: NRL Memorandum Report 2811
2. GOVT ACCESSION NO.:
3. RECIPIENT'S CATALOG NUMBER:
4. TITLE (and Subtitle): VOLUME INTERSECTION OF TWO IDENTICAL RIGHT CIRCULAR CONES
5. TYPE OF REPORT & PERIOD COVERED: Interim report
6. PERFORMING ORG. REPORT NUMBER:
7. AUTHOR(s): R. G. Hughes, J. D. Clamons
8. CONTRACT OR GRANT NUMBER(s):
9. PERFORMING ORGANIZATION NAME AND ADDRESS: Naval Research Laboratory, Washington, D. C. 20375
10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS: NRL Prob. 84G05-10
11. CONTROLLING OFFICE NAME AND ADDRESS:
12. REPORT DATE: June 1974
13. NUMBER OF PAGES: ~~20~~ 18
14. MONITORING AGENCY NAME & ADDRESS (if different from Controlling Office):
15. SECURITY CLASS. (of this report): UNCLASSIFIED
15a. DECLASSIFICATION/DOWNGRADING SCHEDULE:
16. DISTRIBUTION STATEMENT (of this Report): Approved for public release; distribution unlimited.
17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report):
18. SUPPLEMENTARY NOTES:
19. KEY WORDS (Continue on reverse side if necessary and identify by block number):

Doppler Flow Meter
Reverberation Volume

20. ABSTRACT (Continue on reverse side if necessary and identify by block number)

The common volume of two cones representing the reverberation volume of a transducer pair included in a doppler flow meter was calculated. Variables in the calculation, written as a computer program, included the divergence angle of the cones, transducer separation, and transducer size. The volume was found to increase exponentially with an increase in the apex angle, ψ, of the cones for a fixed intersection angle, θ, limited to the condition $\psi < \theta$.

DD FORM 1 JAN 73 1473 EDITION OF 1 NOV 65 IS OBSOLETE

S/N 0102-014-6601

UNCLASSIFIED 18

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

TABLE OF CONTENTS

VOLUME INTERSECTION OF TWO IDENTICAL RIGHT CIRCULAR CONES

Ronald G. Hughes
J. Dean Clamons

INTRODUCTION

The successful operation of an instrument system depends on signal-to-noise levels as well as sensitivity. During the design phase of engineering hardware, a calculation is required to predict the applicability, within a particular set of physical conditions, of the total instrument package to the phenomena to be measured. This provides the constraints or limits, related to sensitivity and signal-to-noise level, which must be met by each of the instrument's system modules, and these serve as the goals for module parameters necessary to produce a functional instrument.

As applied to the doppler current meter, a calculation to determine the volume reverberation signal strength is made, based on the assumption used with sonar frequencies, that the reverberation signal strength is directly proportional to the common intersection volume of the beam patterns of a transmitter-receiver pair[1] (see Fig. 1). Concentric beams result in an infinite volume, excluding consideration of acoustic signal attenuation, but introduce the problem of testing the transducer system in a finite container since the reflection from the

Manuscript submitted May 8, 1974.



Fig. 1 - Two dimensional view of the volume intersection of two identical right circular cones of divergence angle ψ and intersection angle θ.

walls results in a noise signal which far exceeds the volume reverberation. But when the beams of the transducer pair are crossed, the desired signal is observed in the output of the signal processing electronics.

Following the above, the magnitude of the signal expected in the receiving transducer of the doppler meter configuration is obtained as a volume reverberation. This signal strength is assumed directly proportional to the common volume which is viewed by the receiving transducer and insonified by the transmitter, presupposing suppression of side lobes. The predicted value of the backscattered signal is calculated as a product of factors related to this common volume, the volume scattering function model, the beam pattern spreading, and the attenuation loss.

To implement this approach in predicting signal levels, the volume intersection of the doppler meter transducer pair is calculated with an assumption that both the radiated and received beam patterns are conical and identical.

MATHEMATICAL FORMULATION

The calculation of the volume intersection of two right circular cones is very simple in principle, but involves the tedious evaluation of a triple integral. Utilization of the symmetry of the problem in cylindrical coordinates fails since the area formed by the intersection of a cone with a plane not perpendicular to the symmetry axis is elliptical rather than circular.

Further restrictions imposed on the problem follow:

1. The vertices of the two cones are positioned on the x-y plane such that $Z = y = 0$.

2. The vertices are separated by a distance $D > 0$.

Variables include:

θ, the angle of intersection of the symmetry axes of the two beams;

ψ divergence angle of transducer beam (apex angle of the cone);

A1, radius of the transducers; and

S, separation of the centers of the transducers.

The analytic expression for a cone having its axis of symmetry congruent with the Z axis of an orthogonal Cartesian coordinate system is given by

$$x^2/a^2 + y^2/b^2 - \frac{Z^2}{C^2} = 0 \qquad (1)$$

For the right circular case eq. (1) changes from (see Fig. 2) to

$$x^2 + y^2 - C^2Z^2 = 0 \qquad (2)$$

The constant C appearing in equation (2) is defined in terms of the divergence angle, ψ, specified for the transducers, as

$$C = \tan \psi /2.$$

The expression for the rotated and translated cone in terms of the Cartesian coordinate system in Fig. 1 is determined by means of a rotation and translation operation applied to equation (2). The rotation operators about the y axis for angles $-\theta_1$ and $+\theta_1$ for cones 1 and 2 respectively are:



Fig. 2 - Three dimensional plot of the general expression for a right circular cone with vertex at the origin.

$$Ry_1 = \begin{pmatrix} \cos\theta_1 & 0 & -\sin\theta_1 \\ 0 & 1 & 0 \\ \sin\theta_1 & 0 & \cos\theta_1 \end{pmatrix}$$

$$Ry_2 = \begin{pmatrix} \cos\theta_1 & 0 & \sin\theta_1 \\ 0 & 1 & 0 \\ -\sin\theta_1 & 0 & \cos\theta_1 \end{pmatrix}.$$

The angle θ_1 in the expressions for the rotation operators is defined in terms of angle ϕ as noted in Fig. 1: $\theta_1 = \phi - \pi/2$ for cone 1, and $\theta_1 = \pi/2 - \phi$ for cone 2. These substitutions in Ry_1 and Ry_2 result in:

$$Ry_1 = \begin{pmatrix} \sin\phi & 0 & -\cos\phi \\ 0 & 1 & 0 \\ \cos\phi & 0 & \sin\phi \end{pmatrix}$$

$$Ry_2 = \begin{pmatrix} \sin\phi & 0 & \cos\phi \\ 0 & 1 & 0 \\ -\cos\phi & 0 & \sin\phi \end{pmatrix}$$

Operating on equation (2) with Ry_1 and translating along the x axis by -D/2 yields for cone 1:

$$((x + D/2)\sin\phi - Z\cos\phi)^2 + y^2 = \tan^2 \psi/2\, ((x + D/2) \cos\phi + Z\sin\phi)^2 \qquad (3)$$

A similar application of Ry_2 and translation of D/2 on the axis produces the equation for cone 2 as

$$((x - D/2)\sin\phi + Z\cos\phi)^2 + y^2 = \tan^2 \psi/2\,((x - D/2)\cos\phi - Z\sin\phi)^2 \qquad (4)$$

Because of the symmetry of the intersection volume with respect to the $x = 0$ and $y = 0$ planes, it is necessary to consider only that part of volume which lies in the first $(x - y)$ quadrant. The volume under consideration can be further segmented into Volume V_a and V_b as shown in Fig. 1. Simplification of integration results in particular from setting the limits as V_a is bounded by cone 1 and the planes $x = 0$, $y = 0$, $Z = Z_2$ and V_b is bounded by cone 2 and the planes $x = 0$, $y = 0$, $Z = Z_2$. The integral in Cartesian coordinates for Va is found to be:

$$\frac{V_a}{4} = \int_{Z_1}^{Z_2} dZ \int_0^{Z/\tan(\phi - \psi/2) - D/2} dx \int_0^{\left(\tan^2 \psi/2\,((D/2 + x)\cos\phi + Z\sin\phi)^2 - ((x + D/2)\sin\phi - Z\cos\phi)^2\right)^{1/2}} dy$$

Integrating this expression with respect to the y variable and redefining the terms in the resulting integral as

$$A = \sin^2\phi \tan^2 \psi/2 - \cos^2\phi$$

$$B = 2\sin\phi\cos\phi\,(\tan^2 \psi/2 + 1)$$

$$C = \cos^2\phi \tan^2 \psi/2 - \sin^2\phi$$

$$T1 = Z/\tan(\phi - \psi/2)$$

$$f(x) = \sqrt{x^2C + ZBx - Z^2A}$$

results in the following expression:

$$\frac{V_a}{4} = \frac{1}{\sqrt{-C}} \int_{Z_1}^{Z_2} dZ \; \frac{x}{2} \sqrt{\frac{Z^2(B^2 - 2AC)}{-4C} - x^2} - \left(\frac{Z^2(B^2 - 4AC)}{-8C} \right)$$

$$\sin^{-1} \left(\frac{2x\sqrt{-C}}{Z\sqrt{B^2 - 4AC}} \right) \Bigg]_{f(D/2)}^{f(T1)}$$

The same equation yields $V_b/4$ with the substitutions

B	-B
D	-D
T1	$-Z/\tan(\phi + \psi/2)$

and the total volume of intersection is obviously $V = 4(V_a/4 + V_b/4)$.

RESULTS

The intersection volume was calculated for a fixed transducer spacing S and radius A1, for varying beam divergence angle ψ, and beam intersection angle θ and the results are plotted in Fig. 3. Inspection of the plot indicates that the intersection volume differs from an exponential function as the beam intersection angle approaches the beam divergence angle. This result is particularly noticeable for the parameter $\theta = 6.5^\circ$. The plots for the two other beam intersection angles are non-exponential to a lesser extent over the range of values taken for ψ.

In order to check the numerical integration method, a Monte Carlo program was written to calculate the intersection volume for $\theta = 8^\circ$ and a range of ψ from 2° to 4.5°. For this calculation, the transducer

separation was fixed at 3.81 cm and the transducer radii at 0.65 cm. The nearly exponential dependence of the calculated volume as a function of ψ is plotted in Fig. 4. The plot shows the comparison of the numerical integration and Monte Carlo methods to be nearly equivalent given the same beam geometry. The calculated intersection volumes found by these two methods indicate that the results of an earlier study[2] are too liberal in approximating the size of the assumed elliptical cross section.



Fig. 3 - Intersection volume versus divergence angle ψ for transducers of .635 cm radius and 3.81 cm separation center to center.



Fig. 4 - Comparison of the calculated intersection volumes as a function of full beam width.

REFERENCES

1. Physics of Sound in the Sea (Part II Reverberation) Research Analysis Group (Committee on Undersea Warfare) National Research Council.

2. D.G. Polvani, Ocean Engineering Memo #73-04 (11 Jan 1973) Westinghouse Electric Corporation, Annapolis, Md.

APPENDIX I

This appendix includes a copy of the FORTRAN IV coded program which was used to perform the intersection volume calculations. Instructions for operating the program are included in comment statements listed on the printed output. In addition, definitions of the input constants of transducer size, intersection angle, and divergence angle of the acoustic beam are listed.

```
 73/04/13. 11.00.20.
PROGRAM   VOLUME

PROGRAM VOLUME(INPUT,OUTPUT)
*          THIS PROGRAM CALCULATES THE VOLUME OF INTERSECTION OF
*        THE BEAM PATTERNS OF TWO IDENTICAL TRANSDUCERS.  THE BEAM
*        PATTERNS ARE ASSUMED TO BE CONICAL WITH AN INCLUDED ANGLE
*        OF PSI DEGREES, AND THEIR AXES INTERSECT AT AN ANGLE OF
*        THETA DEGREES.  THE RADIUS OF THE TRANSDUCERS IS A1 AND
*        THE SEPARATION BETWEEN TRANSDUCER CENTERS IS S.
*          THE VOLUME IS FOUND BY CALCULATING THE AREA OF THE
*        HORIZONTAL CROSS-SECTION AND NUMERICALLY INTEGRATING THE
*        AREA WITH RESPECT TO THE Z-COORDINATE USING A ROMBERG
*        INTEGRATION SUBROUTINE TO GET THE VOLUME.
*          FOR DERIVATION OF THE FORMULAS AND A MORE DETAILED
*        DESCRIPTION OF THE GEOMETRY SEE:
*
*              INPUTS:    PSI IN DEGREES
*                         THETA IN DEGREES
*                         A1 IN ANY UNITS
*                         S IN THE SAME UNITS AS A1
*             OUTPUTS:    VOLUME IN THE SAME UNITS AS A1
*
*
COMMON TEMP,C,B,A,D,T2
EXTERNAL FCN1
30 PRINT,*INCLUDED ANGLE OF TRANSDUCER BEAM PATTERN (PSI)*
READ,PSI1
PRINT,*ANGLE OF INTERSECTION OF BEAMS (THETA)*
READ,THETA1
PRINT,*RADIUS OF TRANSDUCERS (A1)*
READ,A1
PRINT,*DISTANCE BETWEEN TRANSDUCER CENTERS (S)*
READ,S
*        CALCULATE PHI (THE ANGLE OF TILT OF THE CONES) AND CONVERT
*        ALL ANGLES TO RADIANS.  FOR CALCULATION PURPOSES PSI/2
*        IS USED.
THETA=THETA1*3.14159/180.0
PHI1=(180.0-THETA1)/2.0
PHI=PHI1*3.14159/180.0
PSI=PSI1*3.14159/180.0
PS1=PSI/2.0
*        CALCULATE D (THE DISTANCE BETWEEN VERTICES OF THE CONES)
*        CALCULATE Z (DISTANCE FROM PLANE OF VERTICES TO INTERSECTIO

*        OF AXES.
X=A1*SIN(THETA)/TAN(PSI)
D=2.0*X+S
Z=D*TAN(PHI)/2.0
*        IF PS1+PHI>90 DEGREES VOLUME IS INFINITE
IF(PSI+PHI.GT.3.14159/2.0)GO TO 1
*        CALCULATE LIMITS OF INTEGRATION (Z1,Z2,Z3) AND CONSTANTS
*        USED IN CALCULATION (A,B,C,TEMP,T2)
S1=SIN(PHI)
C1=COS(PHI)
T=TAN(PSI)
T2=TAN(PHI-PSI)
A=S1*T*S1*T-C1*C1
B=2.0*S1*C1*(T*T+1.0)
C=C1*T*C1*T-S1*S1
TEMP=4.0*T*T
Z1=D*T2/2.0
```

```
Z2=D*SIN(PHI+PSI)*SIN(PHI-PSI)/SIN(2.0*PHI)
Z3=D*TAN(PHI+PSI)/2.0
*      CALCULATE VA (LOWER VOLUME)
CALL XCROMB(Z1,Z2,10E-8,VA,IERR,FCN1)
*      CHANGE VARIABLES FOR CALCULATING VB
D=-D
B=-B
T2=-TAN(PHI+PSI)
*      CALCULATE VB (UPPER VOLUME)
CALL XCROMB(Z2,Z3,10E-8,VB,IERR,FCN1)
*      CALCULATE TOTAL VOLUME AND PRINT RESULTS
V=4.0*(VA+VB)
D=-D
PRINT 500,PSI1,THETA1,PHI1,S,A1,D,V
500 FORMAT(*PSI= *,F10.2,*   THETA= *,F10.2,*   PHI= *,F10.2,
+      /*S= *,F10.2,*   A1= *,F10.2,*   D= *,F10.2/
+      *VOLUME OF INTERSECTION = *,E15.8)
GO TO 20
1 PRINT 501,PSI1,THETA1,PHI1,S,A1,D
501 FORMAT(*PSI= *,F10.2,*   THETA= *,F10.2,*   PHI= *,F10.2/
+      *S= *,F10.2,*   A1= *,F10.2,*   D= *,F10.2/
+      *VOLUME OF INTERSECTION IS INFINITE*)
20 PRINT 505
505 FORMAT(//)
GO TO 30
END
FUNCTION FCN1(Z)
*        THIS FUNCTION CALCULATES THE AREA OF CROSS-SECTION AT
*      HEIGHT Z.  IT FIRST CALCULATES THE LIMITS AT WHICH THE
*      AREA INTEGRAL MUST BE EVALUATED AND THEN CALLS FCN2 AT
*      THESE LIMITS TO MAKE THE EVALUATIONS.
COMMON TEMP,C,B,A,D,T2
Y1=D/2.0
Y2=Z/T2
Y1=C*Y1*Y1+Z*Y1*B+Z*Z*A
Y2=C*Y2*Y2+Z*Y2*B+Z*Z*A
IF(ABS(Y1).LT.10E-10)Y1=0.0
IF(ABS(Y2).LT.10E-10)Y2=0.0
IF(Y1.LT.0.0)PRINT,*Y1↑2=*,Y1
IF(Y2.LT.0.0)PRINT,*Y2↑2=*,Y2
Y1=SQRT(ABS(Y1))
Y2=SQRT(ABS(Y2))
FCN1=FCN2(Y2,Z)-FCN2(Y1,Z)
RETURN
END
FUNCTION FCN2(X,Z)
*        THIS FUNCTION EVALUATES THE AREA INTEGRAL AT X,Z.
COMMON TEMP,C,B,A,D,T2
RAD=Z*Z*TEMP/(-4.0*C)
F=X/SQRT(RAD)
F2=RAD-X*X
IF(F.LE.1.0) GO TO 1
PRINT,*ARG OF ASIN > 1*,F
F=1.0
F2=0.0
1 FCN2=(X*SQRT(F2)-RAD*ASIN(F))/(SQRT(-C)*2)
RETURN
END
READY.
```